Modern Assembly Language Programming
with the
ARM processor
Chapter 8: Non-Integral Mathematics
1. Base Conversion of Nonintegral Numbers
2. Fixed Point Calculations
3. Fixed Point Multiplication and Division
4. Fixed Point Sine and Cosine
5. Floating Point
6. Algorithms for Floating Point
7. Algorithms for Floating Point
Converting Fractions in Any Base to Decimal

The value 101.0101\textsubscript{2} can be converted to base ten by expanding it as follows:

\[ 1 \times 2^2 + 0 \times 2^1 + 1 \times 2^0 + 0 \times 2^{-1} + 1 \times 2^{-2} + 0 \times 2^{-3} + 1 \times 2^{-4} \]
\[ = 4 + 0 + 1 + 0 + \frac{1}{4} + 0 + \frac{1}{16} \]
\[ = 5.3125\textsubscript{10} \]

Likewise, the hexadecimal fraction 4F2.9A0 can be converted to base ten by expanding it as follows:

\[ 4 \times 16^2 + 15 \times 16^1 + 2 \times 16^0 + 9 \times 16^{-1} + 10 \times 16^{-2} + 0 \times 16^{-3} \]
\[ = 1024 + 240 + 2 + \frac{9}{16} + \frac{10}{256} + \frac{0}{4096} \]
\[ = 1266.6015625\textsubscript{10} \]
Converting Decimal Fractions to Binary

- The integer part is converted by repeated division (Chapter 1).
- The fractional part is converted by repeated multiplication.

Example: Convert the decimal value 5.625 to a binary representation. The integer part is 101₂.

1. Multiply the decimal fraction by two. The integer part of the result is the first binary digit to the right of the radix point.
   Because \(0.625 \times 2 = 1.25\), the first binary digit to the right of the point is a 1. So far, we have \(0.625_{10} = 0.1₂\)

2. Disregard the integer portion and multiply by 2 once again.
   Because \(0.25 \times 2 = 0.50\), the second binary digit to the right of the point is a 0. So far, we have \(0.625_{10} = 0.10₂\)

3. Disregard the integer portion and multiply by 2 once again.
   Because \(0.50 \times 2 = 1.00\), the third binary digit to the right of the point is a 1. So now we have \(0.625 = 0.101\)

4. Disregard the integer portion. Since that leaves us with 0.0, we know that all remaining digits will be zero.

\[5.625_{10} = 101.101₂\]
Nonterminating (Repeating) Binary Fractions

In Radix Notation there are some fractions that repeat and never terminate. Example: The binary representation of the decimal fraction 1/10 is a non-terminating repeating fraction.

\[ \begin{align*}
.1 \times 2 & = 0.2 \\
.2 \times 2 & = 0.4 \\
.4 \times 2 & = 0.8 \\
.8 \times 2 & = 1.6 \\
.6 \times 2 & = 1.2 \\
.2 \times 2 & = 0.4
\end{align*} \]

The last step is exactly the same as the second step.

If we continue, we will repeat the sequence of steps 2-5 forever.

Hence, the final binary representation will be.

\[ 0.1_{10} = .000110011001100111\ldots_2 \]

\[ 0.1_{10} = .00011_2 \]
Nonterminating Binary Fractions - How Many?

**Question:** Are there, in some sense of the word, more nonterminating decimals than binimals, more nonterminating binimals than decimals, or neither?

**Answer:** Obviously, since both are infinite sets, in one sense of the word, the answer is neither. But that is an oversimplification.

If we ask our question differently, we can discover some important information.

**Question:** Is the set of terminating decimals a subset of the set of terminating binimals, or vice versa, or neither.

**Answer:** The set of terminating binimals is a subset of the set of terminating decimals, but the set of terminating decimals is not a subset of the set of terminating binimals.
Lemma

If $0 < x < 1$, terminates in some base $B$ (a product of primes), then $x = \frac{N_x}{D_x}$, and $D_x = p_1^{k_1} p_2^{k_2} \ldots p_n^{k_n}$, where the $p_i$ are the prime factors of $B$.

Proof.

Let $x = \frac{N_x}{D_x}$, and $D_x = p_1^{k_1} p_2^{k_2} \ldots p_n^{k_n}$, where the $p_i$ are the prime factors of $B$. Then $D_x \mid N_x \times B^{k_{max}}$, where $k_{max} = \max(k_1, k_2, \ldots k_n)$, so $x = \frac{N_x}{D_x x}$ terminates after $k_{max}$ or fewer divisions.

Let $x = \frac{N_x}{D_x}$ terminate after $k$ divisions. Then $D_x \mid N_x \times B^k$. Since $D_x$ does not evenly divide $N_x$, $D_x$ must be composed of some combination of the prime factors of $B$. Thus, $D_x$ can be expressed as $p_1^{k_1} p_2^{k_2} \ldots p_n^{k_n}$. \qed
Theorem

The set of terminating binimals is a subset of the set of terminating decimals, but the set of terminating decimals is not a subset of the set of terminating binimals.

Proof.

Let $b$ be a terminating binimal. Then, by the Lemma on the previous slide, $b = \frac{N_b}{D_b}$, such that $D_b = 2^k$, for some $k \geq 0$. Therefore, $D_b = 2^k5^m$, for some $k, m > 0$, and again by the Lemma, $b$, is also a terminating decimal.
Nonterminating Binary Fractions - Examples

Let \( B = 30 = 2 \times 3 \times 5 \). Then any number with denominator \( 2^{k_1}3^{k_2}5^{k_3} \) terminates in base \( B \).

For example, \( \frac{11}{15} \) will terminate in base 30 after one division since \( 15 = 3^15^1 \).

To see what the number will look like, let’s extend the hexadecimal system of using letters to represent digits beyond 9. So we get this chart for base 30:

|------|------|------|------|------|------|------|------|------|------|

Since \( \frac{11}{15} = \frac{22}{30} \), it is \( 0.\text{M}_{30} \).

What is \( \frac{13}{45} \) in base 30? Since \( 45 = 3^35^1 \), this number will have two digits following the radix point. To compute the value, we will have to raise it to higher terms, using \( 30^2 \) as the denominator:

\[
\frac{13}{45} = \frac{260}{900}.
\]

\( \frac{260}{30} = 8 \) with remainder 20, so \( \frac{13}{45} = 0.\text{K}_{30} \).
Nonterminating Binary Fractions - Bottom Line

Since the set of terminating binimals is a subset of the set of terminating decimals, but not vice versa:

- there are terminating fractions in base 10 which do not terminate in base 2, but
- all fractions that terminate in base 2 will also terminate in base 10.

No number system is immune from this problem. There are fractions in base 30, for example, which do not terminate. Changing the base of the computer’s number system is not the answer.
**Fixed Point Numbers**

- Given a finite number of bits, a computer can only *approximately* represent nonintegral numbers.
- Some computers have no hardware support for nonintegral calculations.
- Real numbers can be represented as *fixed point* numbers.

<table>
<thead>
<tr>
<th>Number</th>
<th>Integer Part</th>
<th>Fractional Part</th>
<th>Radix Point</th>
</tr>
</thead>
<tbody>
<tr>
<td>U(7,9)</td>
<td>0010111100101011</td>
<td></td>
<td></td>
</tr>
<tr>
<td>S(9,6)</td>
<td>0010111100101011</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Sometimes it is useful to represent a negative number of bits in the integer part.

Fixed Point Numbers Continued

U(-8, 16) 0 0 0 0 0 0 0 f f f f f f f f f f f f f f

Radix Point

Fractional Part

S(-7, 16) s s s s s s f f f f f f f f f f f f f f

Radix Point

Sign

Fractional Part
Size and Radix Point

The combination of size and radix will affect several properties.

**Precision:** the maximum number of non-zero bits representable,

**Resolution:** the smallest non-zero magnitude representable,

**Accuracy:** the magnitude of the maximum difference between a true real value and it’s approximate representation,

**Range:** the difference between the greatest and smallest number that can be represented,

**Dynamic range:** the ratio of the maximum absolute value, and the minimum positive absolute value representable.
Size and Radix Point – Example

<table>
<thead>
<tr>
<th>Sign</th>
<th>Radix Point</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
</tr>
</tbody>
</table>

\[ S(9, 6) \]

<table>
<thead>
<tr>
<th>Integer Part</th>
<th>Fractional Part</th>
</tr>
</thead>
<tbody>
<tr>
<td>0010111100101111</td>
<td></td>
</tr>
</tbody>
</table>

**Precision:** 16 bits

**Resolution:** \[ A = 2^{-6} = 0.015625 \]

**Accuracy:** \[ A = \frac{R}{2} = 0.0078125 \]

**Range:** Minimum value is \[ 1111111111.111111 = -1024 \]
Maximum value is \[ 0111111111.111111 = 1023.9921875 \]
Range is \[ 1023.9921875 + 1024 = 2047.9921875 \]

**Dynamic range:** For a signed fixed-point rational representation, \( S(a, b) \), the dynamic range is

\[
2 \times \frac{2^a}{2^{-b}} = 2^{a+b+1} = 2^N \\
2^{16} = 65536
\]
Adding and Subtracting

Fixed point addition and subtraction work exactly like their integer counterparts.

\[
\begin{array}{c}
2.25 \\
+ 1.50 \\
\hline
3.75
\end{array}
\begin{array}{c}
0.010000 \\
+ 0.001100 \\
\hline
0.011100
\end{array}
\]

\[
\begin{array}{c}
11.125 \\
- 5.625 \\
\hline
5.500
\end{array}
\begin{array}{c}
0.101100 \\
+ 0.110100 \\
\hline
0.010100
\end{array}
\]

\[
\begin{array}{c}
-12.375 \\
+ 5.250 \\
- 7.125
\end{array}
\begin{array}{c}
1.001101 \\
+ 0.010101 \\
1.100011
\end{array}
\]

In fact, integer math is just fixed point with no bits in the fractional part.
Multiplication

The result of multiplying an $n$ bit number by an $m$ bit number is an $n + m$ bit number.

\[
\begin{array}{c}
3.75 \\
\times 2.50 \\
\hline
.0000
\end{array}
\quad =
\begin{array}{c}
00011.110 \\
\times 00010.100 \\
\hline
0001.1110
\end{array}
\begin{array}{c}
1.875 \\
\hline
.0000
\end{array}
\begin{array}{c}
7.50 \\
\hline
9.3750
\end{array}
\begin{array}{c}
000111.10 \\
\hline
00000001001.011000
\end{array}
\]

Note that the radix point in the result has shifted.

The result of multiplying two numbers $S(a_1, b_1)$ and $S(a_2, b_2)$ is an $S(a_1 + a_2 + 1, b_1 + b_2)$ number.

Fixed point multiplies may be followed by an appropriate shift.
Fixed Point Multiplication Result Formats

**Unsigned Multiplication**

The result of multiplying two unsigned numbers $U(i_1, f_1)$ and $U(i_2, f_2)$ is a $U(i_1 + i_2, f_1 + f_2)$ number.

**Mixed Multiplication**

The result of multiplying a signed number $S(i_1, f_1)$ and an unsigned number $U(i_2, f_2)$ is an $S(i_1 + i_2, f_1 + f_2)$ number.

**Signed Multiplication**

The result of multiplying two signed numbers $S(i_1, f_1)$ and $S(i_2, f_2)$ is an $S(i_1 + i_2 + 1, f_1 + f_2)$ number.
Fixed Point Multiplication on ARM

@@ Multiply two S(10,5) numbers
mul r0,r1,r2 @ x = a * b -> S(21,10)

@@ Multiply two U(12,4) numbers and produce a U(12,4)
mul r3,r4,r5 @ x = a * b -> U(24,8)
lsr r3,r3,#4 @ shift back to U(12,4)

@@ Multiply two S(16,15) numbers and produce an S(16,15)
smull r0,r1,r2,r3 @ x = a * b -> S(33,30)
lsr r0,r0,#17 @ get 15 bits from r0
orr r0,r1,lsl #15 @ combine with 17 bits from r1

@@ Multiply two U(10,22) numbers
umull r0,r1,r2,r3 @ x = a * b -> U(20,44)
Division

Given a dividend, $N$, with format $\text{U}(i_1,f_1)$ and a divisor, $D$, with format $\text{U}(i_2,f_2)$.

The value of the least significant bit of $N$ is $2^{-f_1}$ and the value of the least significant bit of $D$ is $2^{-f_2}$.

In order to perform the division using integer operations, it is necessary to multiply $N$ by $2^{f_1}$ and multiply $D$ by $2^{f_2}$ so that both numbers are integers. Therefore the division operation can be written as:

$$Q = \frac{N \times 2^{f_1}}{D \times 2^{f_2}} = \frac{N}{D} \times 2^{f_1-f_2}.$$

Example: Given two $\text{U}(5,3)$ numbers:

$$Q = \frac{N \times 2^3}{D \times 2^3} = \frac{N}{D} \times 2^0.$$

If the programmer wants to have $f_r$ fractional bits in the result, then the dividend must be shifted left an additional $f_r$ bits before the division.
Division Example

For example, suppose the programmer wants to divide 01001.011 stored as a U(28,3) by 00011.110 which is also stored as a U(28,3), and wishes to have six fractional bits in the result. The programmer would first shift 01001.011 to the left by six bits, then perform the division and compute the position of the radix in the result as shown:

\[
01001.011 \div 00011.110 = (0000001001011000000 \div 00011110) \times 2^{-6-3+3}
\]

\[
\begin{array}{c}
11110)1001011000000 \\
111100000000 \\
1111000000 \\
1111000000 \\
0
\end{array}
\]

\[
10100000 \times 2^{-6} = 10.100000
\]
Consider:
The largest possible value of the dividend is \( N_{\text{max}} = 2^i_1 - 2^{-f_1} \), and the smallest positive value for the divisor is \( D_{\text{min}} = 2^{-f_2} \). Therefore, the maximum quotient is given by:

\[
Q_{\text{max}} = \frac{2^i_1 - 2^{-f_1}}{2^{-f_2}} = 2^{i_1+f_2} - 2f_1f_2.
\]

Taking the limit of the previous equation,

\[
\lim_{f_1-f_2 \to -\infty} Q_{\text{max}} = 2^{i_1+f_2},
\]

provides the following bound on how many bits are required in the integer part of the quotient:

\[
Q_{\text{max}} < 2^{i_1+f_2}.
\]

Therefore, in the worst case, the quotient will require \( i_1 + f_2 \) integer bits.
Division Result Formats

Unsigned Division
The result of dividing an unsigned fixed point number $U(i_1, f_1)$ by an unsigned number $U(i_2, f_2)$ is a $U(i_1 + f_2, f_1 - f_2)$ number.

Mixed Division
The result of dividing two fixed point numbers where one of them is signed and the other is unsigned is an $S(i_1 + f_2, f_1 - f_2)$ number.

Signed Division
The result of dividing two signed fixed point numbers is an $S(i_1 + f_2 + 1, f_1 - f_2)$ number.
Division Without Losing Precision

The smallest positive value for the dividend is \( N_{\text{min}} = 2^{-f_1} \), and the largest possible value of the divisor is \( D_{\text{max}} = 2^{i_2} - 2^{-f_2} \).

In the worst case, the least significant bit of the quotient will be \( 2^{-(i_2+f_1)} \).

Shifting the dividend left by \( i_2 + f_2 \) bits will convert it into a U(\( i_1, i_2 + f_1 + f_2 \)).

When it is divided by a U(\( i_2, f_2 \)), the result is a U(\( i_1 + f_2, i_2 + f_1 \)). This is the minimum size which is guaranteed to preserve all bits of precision.

The general method for performing fixed point division while maintaining maximum precision is as follows:

1. shift the dividend left by \( i_2 + f_2 \) then

2. perform the division, and

3. use the rules from the previous slide to determine the result format.
Division By a Constant Revisited

Calculate \( x \div 23 \) using only 8-bit signed integer multiplication.

The reciprocal of 23 is

\[
R = \frac{1}{23} = 0.000010110010001011\ldots_2.
\]

If we store \( R \) as an \( S(1,11) \), it would look like this:

![Fractional Part Radix Point Sign](image)

Note that in this format, the reciprocal of 23 has five leading zeros.

We can store more of the significant bits of \( R \) by shifting it left to remove some of the leading zeros.
Division By a Constant Revisited

Radix Point

S(0, 11)

Sign

Fractional Part

0 0 0 0 0 1 0 1 1 0 0 1

Radix Point

S(−1, 11)

Sign

Fractional Part

0 0 0 1 0 1 1 0 0 1

Radix Point

S(−2, 10)

Sign

Fractional Part

0 0 0 1 0 1 1 0 0 1
Division By a Constant Revisited

We cannot shift any more without changing the sign bit.
Division By a Constant Revisited

An $S(7, 0)$ number $x$ multiplied by an $S(-4, 8)$ number $R$ will yield an $S(4, 8)$ number $y$. The value $y$ will be $2^3 \times \frac{x}{23}$ because we have three “hidden” bits to the right of the radix point.

To calculate $y = 101_{10} \div 23_{10}$, we can multiply and perform a shift as follows:

\[
\begin{array}{c}
.01100101 \\
\times 01011010 \\
\hline
0.11001010 \\
011.00101 \\
0110.0101 \\
011001.11 \\
\hline
00100100.00000010
\end{array}
\]

The integer portion, $100011_2$, shifted right three bits, is $100_2 = 4_{10}$. 
Division by a Negative Constant

Calculate \( x \div -9 \) using 8-bit signed integer multiplication.

\[
-\frac{1}{9} = -0.000111000111000111000111\ldots
= 111\ldots1111\ldots
\]

We can represent \( \frac{1}{9} \times 2^2 \) as the following S(7,8) fixed point number:

Note that the upper 5 bits are all one.
Division by a Negative Constant – Part 2

Since the upper 4 bits are all copies of the sign, we don’t need to keep them.

\( \frac{1}{-9} \) in 8 bits, can be represented as an S(\(-3, 8\)) number \( R \):

\[
\begin{array}{cccccccc}
7 & 6 & 5 & 4 & 3 & 2 & 1 & 0 \\
S(-3, 8) & 1 & 1 & 1 & 0 & 0 & 0 & 1 & 1 & 1 & 0 \\
\end{array}
\]

Fractional Part

Given an S(7,0) number \( X \), \( R \times X \) will yield an S(7,8) number, such that

\[
Y \times 2^2 = R \times X = \left( \frac{1}{-9} \times 2^2 \right) \times X
\]

\[
Y = R \times X \times 2^{-2}
\]
Division by a Negative Constant - Example

To calculate $Y = 101_{10} ÷ -9_{10}$, use a signed multiply:

\[
\begin{array}{cccccccccccc}
& . & 1 & 0 & 0 & 0 & 1 & 1 & 1 & 0 \\
\times & 0 & 1 & 0 & 1 & 1 & 0 & 0 & 1 \\
\hline
1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & . & 1 & 0 & 0 & 0 & 1 & 1 & 1 & 0 \\
1 & 1 & 1 & 1 & 1 & 1 & 0 & 0 & . & 0 & 1 & 1 & 1 & 0 \\
1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 & . & 1 & 1 & 1 & 0 \\
1 & 1 & 1 & 0 & 0 & 0 & 1 & 1 & . & 1 & 0 \\
1 & 1 & 0 & 1 & 0 & 1 & 0 & 0 & . & 0 & 1 & 0 & 1 & 1 & 1 & 1 & 0 \\
\end{array}
\]

We can immediately throw away the fractional part of the result, keeping only the upper 8 bits.

$11010100_2$ shifted right 2 bits is: $11110101_2$

$-(00001010 + 1)_2 = -(8 + 2 + 1)_{10} = -11_{10}$.

If the modulus is required, it can be calculated as: $101 - (-11 \times -9) = 2$
Implementing Sine and Cosine

- \( \cos x = \sin \frac{\pi}{2 - x} \)
- \( \sin x \) is cyclical, so \( \sin -2\pi = \sin 0 = \sin 2\pi \ldots \) This means that we can limit the domain of our function to the range \([-\pi, \pi]\)
- \( \sin x \) is symmetric, so that \( \sin -x = -\sin x \). This means that we can further restrict the domain to \([0, \pi]\).
- After we restrict the domain to \([0, \pi]\), we notice another symmetry, \( \sin x = \sin(\pi - x), \frac{\pi}{2} \leq x \leq \pi \) and we can further restrict the domain to \([0, \frac{\pi}{2}]\).
- the range of both functions, \( \sin x \) and \( \cos x \), is in the range \([-1, 1]\).

Write a single shared function, \( \text{sinq} \), to be used by both sine and cosine.
- \( \text{sinq} \) will accept \( x \) as an S(1,30), and
- \( \text{sinq} \) will return an S(1,30)
Taylor Series

\[
\sin x = \sum_{n=0}^{\infty} \frac{(-1)^n x^{2n+1}}{(2n+1)!}
\]

- Only need a few terms to get 30 fractional bits of precision.
- All of the factorial terms are constants.
- Combine the \(-1\) term with the factorials.

\[
\sin x = \sum_{n=0}^{\infty} \frac{-1^n x^{2n+1}}{(2n+1)!}
\]

\[
= \frac{1}{1!} x + \frac{-1}{3!} x^3 + \frac{1}{5!} x^5 + \frac{-1}{7!} x^7 + \ldots
\]

Pre-compute the constants and store them in a table.

\[
= c_0 x + c_1 x^3 + c_2 x^5 + c_3 x^7 + \ldots
\]
Fixed Point Formats

Each successive power-of-$x$ term must be reduced to 32-bits, or the format would become unmanageable:

<table>
<thead>
<tr>
<th>Term</th>
<th>Format</th>
<th>32-bit</th>
</tr>
</thead>
<tbody>
<tr>
<td>$x$</td>
<td>$S(1,30)$</td>
<td>$S(1,30)$</td>
</tr>
<tr>
<td>$x^3$</td>
<td>$S(3,90)$</td>
<td>$S(3,28)$</td>
</tr>
<tr>
<td>$x^5$</td>
<td>$S(5,150)$</td>
<td>$S(5,26)$</td>
</tr>
<tr>
<td>$x^7$</td>
<td>$S(7,210)$</td>
<td>$S(7,24)$</td>
</tr>
<tr>
<td>$x^9$</td>
<td>$S(9,270)$</td>
<td>$S(9,22)$</td>
</tr>
<tr>
<td>$x^{11}$</td>
<td>$S(11,330)$</td>
<td>$S(11,20)$</td>
</tr>
<tr>
<td>$x^{13}$</td>
<td>$S(13,390)$</td>
<td>$S(13,18)$</td>
</tr>
<tr>
<td>$x^{15}$</td>
<td>$S(15,450)$</td>
<td>$S(15,16)$</td>
</tr>
<tr>
<td>$x^{17}$</td>
<td>$S(17,510)$</td>
<td>$S(17,14)$</td>
</tr>
</tbody>
</table>

- Compute $x^2$ at beginning of function.
- For each successive term:
  - Multiply the previous term by $x^2$ (64-bit result)
  - Discard the lower 32 bits.
Constant Terms

\[-\frac{1}{3!} = \frac{1}{6}\]

Convert to binary:

<table>
<thead>
<tr>
<th>Multiplication</th>
<th>Result</th>
</tr>
</thead>
<tbody>
<tr>
<td>$\frac{1}{6} \times 2 = \frac{2}{6}$</td>
<td>Integer $0$</td>
</tr>
<tr>
<td>$\frac{2}{6} \times 2 = \frac{4}{6}$</td>
<td>Integer $0$</td>
</tr>
<tr>
<td>$\frac{4}{6} \times 2 = \frac{8}{6}$</td>
<td>Integer $1$</td>
</tr>
<tr>
<td>$\frac{2}{6} \times 2 = \frac{4}{6}$</td>
<td>Integer $0$</td>
</tr>
<tr>
<td>$\frac{8}{6} \times 2 = \frac{8}{6}$</td>
<td>Integer $1$</td>
</tr>
</tbody>
</table>

\[-\frac{1}{3!} = -0.00\overline{1}_2\]

\[= \ldots111.1\overline{0}_2\]
Constant Terms (Part 2)

\[-\frac{1}{3!} = -0.0012\]

\[= \ldots 111.110_2\]
<table>
<thead>
<tr>
<th>Term</th>
<th>Reciprocal</th>
<th>Reciprocal</th>
</tr>
</thead>
<tbody>
<tr>
<td>$-\frac{1}{3!}$</td>
<td>S($-2,32$)</td>
<td>AAAAAAAAAA</td>
</tr>
<tr>
<td>$\frac{1}{5!}$</td>
<td>S($-6,32$)</td>
<td>444444445</td>
</tr>
<tr>
<td>$-\frac{1}{7!}$</td>
<td>S($-12,32$)</td>
<td>97F97F97</td>
</tr>
<tr>
<td>$\frac{1}{9!}$</td>
<td>S($-18,32$)</td>
<td>5C778E96</td>
</tr>
<tr>
<td>$-\frac{1}{11!}$</td>
<td>S($-25,32$)</td>
<td>9466EA60</td>
</tr>
<tr>
<td>$\frac{1}{13!}$</td>
<td>S($-32,32$)</td>
<td>5849184F</td>
</tr>
<tr>
<td>$-\frac{1}{15!}$</td>
<td>S($-40,32$)</td>
<td>94603063</td>
</tr>
<tr>
<td>$\frac{1}{17!}$</td>
<td>S($-48,32$)</td>
<td>654B1DC1</td>
</tr>
</tbody>
</table>
Computing Each Term

<table>
<thead>
<tr>
<th>Term</th>
<th>Numerator</th>
<th>Reciprocal</th>
<th>Result</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Value</td>
<td>Format</td>
<td>Value</td>
</tr>
<tr>
<td>1</td>
<td>$x$</td>
<td>S(1,30)</td>
<td>Extend to 64 bits and shift right</td>
</tr>
<tr>
<td>2</td>
<td>$x^3$</td>
<td>S(3,28)</td>
<td>$-\frac{1}{3!}$</td>
</tr>
<tr>
<td>3</td>
<td>$x^5$</td>
<td>S(5,26)</td>
<td>$\frac{1}{5!}$</td>
</tr>
<tr>
<td>4</td>
<td>$x^7$</td>
<td>S(7,24)</td>
<td>$-\frac{1}{7!}$</td>
</tr>
<tr>
<td>5</td>
<td>$x^9$</td>
<td>S(9,22)</td>
<td>$\frac{1}{9!}$</td>
</tr>
<tr>
<td>6</td>
<td>$x^{11}$</td>
<td>S(11,20)</td>
<td>$-\frac{1}{11!}$</td>
</tr>
<tr>
<td>7</td>
<td>$x^{13}$</td>
<td>S(13,18)</td>
<td>$\frac{1}{13!}$</td>
</tr>
<tr>
<td>8</td>
<td>$x^{15}$</td>
<td>S(15,16)</td>
<td>$-\frac{1}{15!}$</td>
</tr>
<tr>
<td>9</td>
<td>$x^{17}$</td>
<td>S(17,14)</td>
<td>$\frac{1}{17!}$</td>
</tr>
</tbody>
</table>
Correcting Shifts for Each Term

<table>
<thead>
<tr>
<th>Term Number</th>
<th>Original Format</th>
<th>Shift Amount</th>
<th>Resulting Format</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>S(1,30)</td>
<td>1</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>2</td>
<td>S(2,61)</td>
<td>0</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>3</td>
<td>S(0,63)</td>
<td>2</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>4</td>
<td>S(−4,64)</td>
<td>6</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>5</td>
<td>S(−8,64)</td>
<td>10</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>6</td>
<td>S(−13,64)</td>
<td>15</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>7</td>
<td>S(−18,64)</td>
<td>20</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>8</td>
<td>S(−24,64)</td>
<td>26</td>
<td>S(2,61)</td>
</tr>
<tr>
<td>9</td>
<td>S(−30,64)</td>
<td>32</td>
<td>S(2,61)</td>
</tr>
</tbody>
</table>
C Implementation

```c
#define pi 0x3243F6A8 /* pi as an S(3,28) */
#define pi_2 0x1921FB54 /* pi/2 as an S(3,28) */
#define pi_x2 0x6487ED51 /* 2*pi as an S(3,28) */

#define TABSIZE 6 /* use first 7 terms of Taylor series */
struct tabentry{int coeff; int shift;};
const static struct tabentry sintab[TABSIZE]= {
    {0xAABBBBBA, 0}, {0x44444444, 2}, {0x97F97F97, 6},
    {0x5C778E96, 10}, {0x9466EA60, 15}, {0x5849184F, 20}};

/* sinq does all of the real work. x is S(1,30) */
static int sinq(int x)
{
    long long sum = (long long)x << 31; /* initialize running total S(2,61) */
    long long xsq = ((long long)x*(long long)x)>>31; /* calculate x^2 S(2,28) */
    long long curpower = x; /* curpower holds x^(2n+1) S(1,30) */
    long long tmp; /* tmp holds each term as it is computed S(2,61) */
    int i=0;
    do {
        curpower = ((curpower * xsq) >> 31); /* calculate x^(2n+1) S(1,30) */
        tmp = curpower * sintab[i].coeff; /* calculate term */
        if(tmp < 0) /* adjust for round-off */
            tmp++; /* if term is negative */
        tmp >>= sintab[i].shift; /* shift to align radix S(2,61) */
        sum += tmp; /* add to running total S(2,61) */
    } while(++i<TABSIZE); /* repeat 5 more times */
    return (sum >> 33); /* shift and truncate to S(3,28) */
}
```
C Implementation

```c
/* fixed_sin_C(S(3,28) x) calculates sine of x*/
int fixed_sin_C(int x)
{
    while(x<0) x += pi_x2;
    while(x>pi_x2) x -= pi_x2;
    if(x<=pi_2) return sinq(x<<2);
    if(x<=pi) return sinq((pi-x)<<2);
    if(x<=(pi+pi_2)) return -sinq((x-pi)<<2);
    return -sinq((pi_x2 -x)<<2);
}

/* fixed_cos_C(S(3,28) x) calculates cosine of x */
int fixed_cos_C(int x)
{
    while(x<0) x += pi_x2;
    x = pi_2 - x;
    return fixed_sin(x);
}
```
@ Name: sincos.S
@ Author: Larry Pyeatt
@ Date: 2/22/2014

This is a version of the sin/cos functions that uses symmetry to enhance precision. The actual sin and cos routines convert the input to lie in the range 0 to pi/2, then pass it to the worker routine that computes the result. The result is then converted back to correspond with the original input.

We calculate sin(x) using the first seven terms of the Taylor Series: sin(x) = x - x^3/3! + x^5/5! - x^7/7! + x^9/9! - ... and we calculate cos(x) using the relationship: cos(x) = sin(pi/2-x).

We start by defining a helper function, which we call sinq. The sinq function calculates sin(x) for 0=x=pi/2. The input, x, must be an S(1,30) number. The factors of x that sinq will use are: x, x^3, x^5, x^7, x^9, x^11, and x^13. Dividing by (2n+1)! is changed to a multiply by a coefficient as we compute each term, we will add it to the sum, stored as an S(2,61). Therefore, we want the product of each power of x and its coefficient to be converted to an S(2,61) for the add. It turns out that this just requires a small shift.

We build a table to decide how much to shift each product before adding it to the total. x^2 will be stored as an...
Assembly Implementation

```nasm
@@ S(2,29), and x is given as an S(1,30). After multiplying
@@ x by x^2, we will shift left one bit, so the procedure is:
@@ x will be an S(1,30) - multiply by x^2 and shift left
@@ x^2 will be an S(3,28) - multiply by x^2 and shift left
@@ x^4 will be an S(5,26) - multiply by x^2 and shift left
@@ x^8 will be an S(7,24) - multiply by x^2 and shift left
@@ x^16 will be an S(9,22) - multiply by x^2 and shift left
@@ x^32 will be an S(11,20)- multiply by x^2 and shift left
@@ x^64 will be an S(13,18)- multiply by x^2 and shift left
@@
@@ The following table shows the constant coefficients
@@ needed for calculating each term.
@@ -1/3! = AAAAAAAA as an S(-2,32)
@@ 1/5! = 44444445 as an S(-6,32)
@@ -1/7! = 97F97F97 as an S(-12,32)
@@ 1/9! = 5C778E96 as an S(-18,32)
@@ -1/11! = 9466EA60 as an S(-25,32)
@@ 1/13! = 5849184F as an S(-32,32)
@@
@@ Combining the two tables of power and coefficient formats,
@@ we can now determine how much shift we need after each
@@ step in order to do all sums in S(2,61) format:
@@ power powerfmt coef coefffmt resultfmt right shift
@@ x S(1,30) * 1 (skip the multiply) 1 -> S(2,61)
@@ x^3 S(3,28) * -1/3! S(-2,32) = S(2,61) 0 -> S(2,61)
@@ x^5 S(5,26) * 1/5! S(-6,32) = S(0,63) 2 -> S(2,61)
@@ x^7 S(7,24) * -1/7! S(-12,32) = S(-4,64) 6 -> S(2,61)
```
Assembly Implementation

@@ \( x^9 \) \( S(9,22) \ast \frac{1}{9!} \) \( S(-18,32) = S(-8,64) \) 10\( \rightarrow \) \( S(2,61) \)

@@ \( x^{11} \) \( S(11,20) \ast \frac{-1}{11!} \) \( S(-25,32) = S(-13,64) \) 15\( \rightarrow \) \( S(2,61) \)

@@ \( x^{13} \) \( S(13,18) \ast \frac{1}{13!} \) \( S(-32,32) = S(-18,64) \) 20\( \rightarrow \) \( S(2,61) \)

.data
.align 2
@@ We will define a few constants that may be useful
.globl pi
pi: .word 0x3243F6A8 @ pi as an \( S(3,28) \)
.globl pi_2
pi_2: .word 0x1921FB54 @ pi/2 as an \( S(3,28) \)
.globl pi_x2
pi_x2: .word 0x6487ED51 @ 2\( \times \)pi as an \( S(3,28) \)

sintab: @@ This is the table of coefficients and shifts
.word 0xA5555555, 0 @ \(-1/3!\) as an \( S(-2,32) \)
.word 0x44444444, 2 @ \(1/5!\) as an \( S(-6,32) \)
.word 0x97F97F97, 6 @ \(-1/7!\) as an \( S(-12,32) \)
.word 0x5C778E96, 10 @ \(1/9!\) as an \( S(-18,32) \)
.word 0x9466EA60, 15 @ \(-1/11!\) as an \( S(-25,32) \)
.word 0x5849184F, 20 @ \(1/13!\) as an \( S(-32,32) \)
.equ tablen,(.-sintab) @ set tablen to the size of table.
@@ The '.' refers to the current address counter value.
@@ Subtracting the address of sintab from the current
@@ address gives the size of the table.
Assembly Implementation

```assembly
.text
@@---------------------------------------------------------
@@ sinq(x)
@@ input: x -> S(1,30) s.t. 0 <= x <= pi/2
@@ returns sin(x) -> S(1,30)

sinq:
    stmfd sp!,{r4-r11,lr}
    smull r2,r4,r0,r0 @ r4 will hold x^2.
    @ @ The first term in the Taylor series is simply x, so
    @ @ convert x to an S(2,61) by doing an asr in 64 bits,
    @ @ and use it to initialize the sum.
    mov r10, r0,lsl #31 @ low 32 bits of sum
    mov r11, r0,asr #1 @ high 32 bits of sum
    @ @ r11:r10 now contains the sum (currently x) as an S(2,61)
    @ @ We are going to convert x^2 to an S(2,28), and round it
    adds r2,r2,#0x40000000 @ Round x^2 up by adding 1 to
        @ the first bit that will be lost.
    adccs r4,r4,#0 @ Propagate the carry.
    lsl r4,r4,#1 @ Make room for one bit in LSB
    orr r4,r4,r2,lsr#31 @ Copy least significant bit of x^2
    @ @ r4 now contains x^2 as an S(2,28)
    mov r5,r0 @ r5 will keep x^(2n-1).
    @ @ r5 now contains x as an S(1,30)
    @ @ The multiply will take time, and on some processors,
    @ @ there is an extra clock cycle penalty if the next
    @ @ instruction requires the result, so do the multiply now.
    smull r0,r5,r4,r5 @ r5:r0 <- x^(2n+1) as an S(4,59)
    ldr r6, =sintab @ get pointer to beginning of table
```
Assembly Implementation

```
add    r7, r6, #tablen @ get pointer to end of table
@@ We know that we will always execute the loop 6 times,
@@ so we use a post-test loop.

sloop:  ldmia r6!,{r8,r9} @ Load two values from the table
@@ r8 now has (-1^n)/(2n+1)!
@@ r9 contains the correcting shift
@@ the previous smull r0,r5,r4,r5 should be complete soon
lsl    r5,r5,#1 @ Shift and copy the m.s. bit of the
orr    r5,r5,r0,lsr#31 @ LSW to l.s. bit of MSW -> S(3,60)
@@ r5 now contains x^(2n+1) as an S(3,60)
@@ Start next multiply now
smull   r0,r1,r5,r8 @ multiply by reciprocal that we
@@ Apply correcting right shift to make an S(2,61).
@@ Note: r9 was loaded from the table earlier.
rsb     r2,r9,#32 @ calculate inverse shift amount
lsr     r0,r0,r9 @ Make room in low word for bits
orr     r0,r0,r1,ls r2 @ paste bits into low word
asr     r1,r1,r9 @ shift upper word right
@@ accumulate result in r10:r11
adds    r10,r10,r0
adc      r11,r11,r1
@@ check to see if there is another term to compute
cmp     r6, r7
@@ Start next multiply now
smulllt r0,r5,r4,r5 @ r5:r0 <- x^(2n+1) as an S(4,59)
@@ The multiply will take three cycles, so start it now
```
blt    sloop    @ Repeat for every table entry
@@ shift result left 1 bit and move MSW to r0
lsl    r11,r11,#1
orr    r0,r11,r10,lsr #31
@@ return the result
ldmfd  sp!,{r4-r11,pc}

@@---------------------------------------------------------
@@ cos(x) NOTE: The cos(x) function does not return.
@@ It is an alternate entry point to sin(x).
@@ input: x -> S(3,28)
@@ returns cos(x) -> S(3,28)
.global fixed_cos
fixed_cos:
  ldr    r1,=pi_x2   @ load pointer to 2*pi
  ldr    r1,[r1]    @ load 2*pi
  cmp    r0,#0      @ Add 2*pi to x if needed, to make
  addle  r0,r0,r1    @ sure x does not become too small
  cosgood: ldr    r1,=pi_2  @ load pointer to pi/2
    ldr    r1,[r1]    @ load pi/2
    sub    r0,r1,r0  @ cos(x) = sin(pi/2-x)
@@ now we just fall through into the sin function
Assembly Implementation

@@---------------------------------------------------------
@@ sin(x)
@@ input: x -> S(3,28)
@@ returns sin(x) -> S(3,28)

.global fixed_sin

fixed_sin:

    stmfd  sp!,{lr}

ldr    r1,=pi_2 @ r1 has pointer to pi/2
ldr    r2,=pi @ r2 has pointer to pi
ldr    r3,=pi_x2 @ r3 has pointer to pi*2

ldr    r1,[r1] @ r1 has pi/2
ldr    r2,[r2] @ r2 has pi
ldr    r3,[r3] @ r3 has pi*2

@@ step 1: make sure x>=0.0 and x<=2pi

negl:  cmp  r0,#0 @ while(x < 0)
       addlt r0,r0,r3 @ x = x + 2 * pi
       blt  negl    @ end while

nonneg: cmp  r0,r3 @ while(x > pi/2)
       subgt r0,r0,r3 @ x = x - 2 * pi
       bgt  nonneg  @ end while

@@ step 2: find the quadrant and call sinq appropriately

inrange: cmp  r0,r1
           blt  chkq2
@@ it is in the first quadrant... just shift and call sinq

1sl  r0,r0,#2
bl  sinq
b  sin_done
Assembly Implementation

```assembly
chkq2:  cmp    r0,r2
       bgt    chkq3
@@ it is in the second quadrant... mirror, shift, and call
@@ sinq
   sub    r0,r2,r0
   lsl    r0,r0,#2
   bl     sinq
   b      sin_done
chkq3:  add    r1,r1,r2    @ we will not need pi/2 again
       cmp    r0,r1    @ so use r1 to calculate 3pi/2
       bgt    chkq4
@@ it is in the third quadrant... rotate, shift, call sinq, @
@@ then complement the result
   sub    r0,r0,r2
   lsl    r0,r0,#2
   bl     sinq
   rsb    r0,r0,#0
   b      sin_done
@@ it is in the fourth quadrant... rotate, mirror, shift, @
@@ call sinq, then complement the result
chkq4:  sub    r0,r0,r2
       sub    r0,r2,r0
       lsl    r0,r0,#2
       bl     sinq
       rsb    r0,r0,#0
```
Assembly Implementation

```assembly
sin_done:
@@ shift result right 2 bits
asr    r0, r0, #2
@@ return the result
ldmfd  sp!, {pc}
@@------------------------------------------
```
# Performance

<table>
<thead>
<tr>
<th>Optimization</th>
<th>Implementation</th>
<th>CPU seconds</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>None</strong></td>
<td>32-bit Fixed Point Assembly</td>
<td>3.85</td>
</tr>
<tr>
<td></td>
<td>32-bit Fixed Point C</td>
<td>18.99</td>
</tr>
<tr>
<td></td>
<td>Single Precision Software Float C</td>
<td>56.69</td>
</tr>
<tr>
<td></td>
<td>Double Precision Software Float C</td>
<td>55.95</td>
</tr>
<tr>
<td></td>
<td>Single Precision VFP C</td>
<td>11.60</td>
</tr>
<tr>
<td></td>
<td>Double Precision VFP C</td>
<td>11.48</td>
</tr>
<tr>
<td><strong>Full</strong></td>
<td>32-bit Fixed Point Assembly</td>
<td>3.22</td>
</tr>
<tr>
<td></td>
<td>32-bit Fixed Point C</td>
<td>5.02</td>
</tr>
<tr>
<td></td>
<td>Single Precision Software Float C</td>
<td>20.53</td>
</tr>
<tr>
<td></td>
<td>Double Precision Software Float C</td>
<td>54.51</td>
</tr>
<tr>
<td></td>
<td>Single Precision VFP C</td>
<td>3.70</td>
</tr>
<tr>
<td></td>
<td>Double Precision VFP C</td>
<td>11.08</td>
</tr>
</tbody>
</table>
Patriot Missile Failure

- On February 25, 1991, during the Gulf War, an American Patriot Missile battery in Dharan, Saudi Arabia, failed to track and intercept an incoming Iraqi Scud missile.

- The Scud struck an American Army barracks, killing 28 soldiers and injuring around 98 other people.

- The cause was an inaccurate calculation of the time since boot. Arithmetic errors.

- The time in tenths of second as measured by the system’s internal clock was multiplied by $\frac{1}{10}$ to produce the time in seconds.

- This calculation was performed using a U(0,24) fixed point representation.

- The small error, when multiplied by the large number giving the time in tenths of a second, led to a significant error.
The Patriot battery had been up for around 100 hours, and an easy calculation shows that the resulting time error due to the magnified chopping error was about 0.34 seconds.

The binary expansion of $\frac{1}{10}$ is 0.00011

The 24 bit register in the Patriot stored 0.00011001100110011001100 introducing an error of 0.0000000000000000000000011002 or about 0.000000095

Multiplying by the number of tenths of a second in 100 hours gives $0.000000095 \times 100 \times 60 \times 60 \times 10 = 0.34$.

A Scud travels at about 1,676 meters per second, and so travels more than half a kilometer in 0.34 seconds.

This was far enough that the incoming Scud was outside the “range gate” that the Patriot tracked.

PEOPLE DIED
Floating Point – Half Precision

Sometimes we need more range than we can get from fixed precision.

IEEE has set standards for various floating point formats.

### IEEE 754 Half-Precision

<table>
<thead>
<tr>
<th>15</th>
<th>14</th>
<th>13</th>
<th>12</th>
<th>11</th>
<th>10</th>
<th>9</th>
<th>8</th>
<th>7</th>
<th>6</th>
<th>5</th>
<th>4</th>
<th>3</th>
<th>2</th>
<th>1</th>
<th>0</th>
</tr>
</thead>
<tbody>
<tr>
<td>s</td>
<td>e_4</td>
<td>e_3</td>
<td>e_2</td>
<td>e_1</td>
<td>e_0</td>
<td>m_9</td>
<td>m_8</td>
<td>m_7</td>
<td>m_6</td>
<td>m_5</td>
<td>m_4</td>
<td>m_3</td>
<td>m_2</td>
<td>m_1</td>
<td>m_0</td>
</tr>
</tbody>
</table>

- The Significand (a.k.a. “Mantissa” or “Fractional Part”) is in sign-magnitude coding, with bit 15 being the sign bit.

- There are 10 bits of significand, but there are 11 bits of significand precision. There is a “hidden” bit, $m_{10}$, between $m_9$ and $e_0$.

- The exponent is an excess-15 number. i.e. The number stored is 15 greater than the actual exponent.
Floating Point – Half Precision Examples

**IEEE 754 Half-Precision**

```
 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1 0
 0 0 1 0 1 0 1 0 0 0 1 0 1 0 1 1
```

- **Sign**: 0 (Positive)
- **Exponent**: 1010 (10 in decimal)
- **Significand**: 011000110001011

\[ +1.10001011 \times 2^{1010} = 1.10001011 \times 2^{10} = 0.000110001011 \]
\[ \approx 0.04819_{10} \]

```
 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1 0
 1 1 1 0 0 0 0 0 0 0 1 0 0 1 0 1
```

- **Sign**: 1 (Negative)
- **Exponent**: 1000 (5 in decimal)
- **Significand**: 0000100101

\[ -1.00001001 \times 2^{1000} = -1.00001001 \times 2^9 = -1000010010.1 \]
\[ = -530.5_{10} \]
Floating Point – Half Precision Special Values

The exponents 00000₂ and 11111₂ have special meaning.

<table>
<thead>
<tr>
<th>Exponent</th>
<th>Significand = 0</th>
<th>Significand ≠ 0</th>
<th>Equation</th>
</tr>
</thead>
<tbody>
<tr>
<td>00000</td>
<td>±0</td>
<td>subnormal</td>
<td>$-1^{sign} \times 2^{-14} \times 0.\text{significand}$</td>
</tr>
<tr>
<td>00001, ..., 11110</td>
<td>normalized value</td>
<td></td>
<td>$-1^{sign} \times 2^{\text{exponent} - 15} \times 1.\text{significand}$</td>
</tr>
<tr>
<td>11111</td>
<td>±∞</td>
<td>NaN</td>
<td></td>
</tr>
</tbody>
</table>

Subnormal means that the value is too close to zero to be completely normalized.

The minimum strictly positive (subnormal) value is $2^{-24} \approx 5.96 \times 10^{-8}$.

The minimum positive normal value is $2^{-14} \approx 6.10 \times 10^{-5}$.

The maximum exactly representable value is $(2 - 2^{-10}) \times 2^{15} = 65504$. 
Floating Point – More Precision

IEEE 754 Single-Precision

31 30 29 28 27 26 25 24 23 22 21 20 . . . 2 1 0

| s | e7 | e6 | e5 | e4 | e3 | e2 | e1 | e0 | m22 | m21 | m20 | ... | m2 | m1 | m0 |

IEEE 754 Double-Precision

63 62 61 60 . . . 54 53 52 51 50 49 48 . . . 2 1 0

| s | e9 | e8 | e7 | ... | e2 | e1 | e0 | m51 | m50 | m49 | m48 | ... | m2 | m1 | m0 |

IEEE 754 Quad-Precision

127 126 125 124 . . . 114 113 112 111 110 109 108 . . . 2 1 0

| s | e13 | e12 | e11 | ... | e3 | e2 | e1 | m111 | m110 | m109 | m108 | ... | m2 | m1 | m0 |

Many processors do not have hardware support for floating point.
Floating Point Addition and Subtraction

The basic algorithm is the same for floats of all sizes.

1. Extract the exponents $E_a$ and $E_b$.
2. Extract the significands $M_a$ and $M_b$. and convert them into 2’s complement numbers, using the signs $S_a$ and $S_b$.
3. Shift the significand with the smaller exponent right by $|E_a - E_b|$.
4. Perform addition (or subtraction) on the significands to get the significand of the result, $M_r$. Remember that the result may require one more significant bit to avoid overflow.
5. If $M_r$ is negative, then take the 2’s complement and set $S_r$ to 1. Otherwise set $S_r$ to 0.
6. Shift $M_r$ until the leftmost 1 is in the “hidden” bit position, and add the shift amount to the smaller of the two exponents to form the new exponent $E_r$.
7. Combine the sign $S_r$, the exponent $E_r$, and significand $M_r$ to form the result.
Floating Point Multiplication and Division

The basic algorithm is the same for floats of all sizes.

1. Calculate the sign of the result $S_r$.
2. Extract the exponents $E_a$ and $E_b$.
3. Extract the significands $M_a$ and $M_b$.
4. Multiply (or divide) the significands to form $M_r$.
5. Add (or subtract) the exponents (in excess-N) to get $E_r$.
6. Shift $M_r$ until the leftmost 1 is in the “hidden” bit position, and add the shift amount to $E_r$.
7. Combine the sign $S$, the exponent $E_r$, and significand $M_r$ to form the result.
Summary

- The two common ways to represent non-integral numbers in a computer are
  - fixed point and
  - floating point.

- Fixed point is a way to perform calculations on non-integral numbers using only integer operations.

- Floating point allows the radix point to be tracked automatically, but requires much more complex software and/or hardware.

- Fixed point will usually provide better performance than floating point, but requires more programming skill.

- Fractions which terminate in base two will also terminate in base ten, but the converse is not true.

- Programmers should avoid counting using fractions which do not terminate in base two, because it leads to the accumulation of round-off errors.